# Maintainer: Massimiliano Torromeo <massimiliano.torromeo@gmail.com>

# ALARM: Kevin Mihelich <kevin@archlinuxarm.org>
#  - remove makedepend on pandoc
#  - patch to remove manual generation that uses pandoc
#  - comment manual installation

pkgname=geoipupdate
pkgver=4.7.1
pkgrel=1
pkgdesc="Update GeoIP2 and GeoIP Legacy binary databases from MaxMind"
license=('Apache' 'MIT')
arch=('x86_64')
url="https://dev.maxmind.com/geoip/geoipupdate"
depends=(glibc)
makedepends=(go)
backup=(etc/GeoIP.conf)
provides=(geoip2-database)
replaces=(geoip2-database)

source=(
    "https://github.com/maxmind/geoipupdate/archive/v$pkgver/$pkgname-$pkgver.tar.gz"
    "geoipupdate.service"
    "geoipupdate.timer"
    "0001-remove-manuals.patch"
)

sha256sums=('e3176325f916571244562a2fec58e0c9ce624f84dbf946616921be8521cbfd94'
            '94d120a089524b91b2c3095332dee66b346bc97f1496cbff677ff02afa37a6cc'
            'ba9039ae9cc3dea4fe48480527b515cab2ad3a2f69aea5bf55f551e6895779e3'
            'bac4deced8219f56e1b394986f5125a6b36864bb5e9ebf0499ec0edcba55f9ed')

prepare() {
    cd "$srcdir/$pkgname-$pkgver"
    sed -i 's/EditionIDs .*/EditionIDs GeoLite2-ASN GeoLite2-City GeoLite2-Country/' conf/GeoIP.conf.default
    patch -p1 -i ../0001-remove-manuals.patch
}

build() {
    cd "$srcdir/$pkgname-$pkgver"

    export CONFFILE=/etc/GeoIP.conf
    export DATADIR=/var/lib/GeoIP

    make BUILDDIR="$srcdir/build" data

    cd cmd/geoipupdate
    go build \
        -trimpath \
        -buildmode=pie \
        -mod=readonly \
        -modcacherw \
        -ldflags "-extldflags \"$LDFLAGS\" -X main.defaultConfigFile=$CONFFILE -X main.defaultDatabaseDirectory=$DATADIR -X main.version=$pkgver" \
        -o "$srcdir/build/geoipupdate" \
        .
}

package() {
    install -dm755 "$pkgdir/var/lib/GeoIP"

    cd "$srcdir"
    install -Dm644 geoipupdate.service "$pkgdir/usr/lib/systemd/system/geoipupdate.service"
    install -Dm644 geoipupdate.timer "$pkgdir/usr/lib/systemd/system/geoipupdate.timer"

    cd "$srcdir/build"
    install -Dm755 geoipupdate "$pkgdir/usr/bin/geoipupdate"
    install -Dm600 GeoIP.conf "$pkgdir/etc/GeoIP.conf"
    install -Dm644 geoipupdate.md "$pkgdir/usr/share/doc/$pkgname/geoipupdate.md"
    install -Dm644 GeoIP.conf.md "$pkgdir/usr/share/doc/$pkgname/GeoIP.conf.md"
    #install -Dm644 geoipupdate.1 "$pkgdir/usr/share/man/man1/geoipupdate.1"
    #install -Dm644 GeoIP.conf.5 "$pkgdir/usr/share/man/man5/GeoIP.conf.5"

    cd "$srcdir/$pkgname-$pkgver"
    install -Dm644 LICENSE-APACHE "$pkgdir/usr/share/licenses/$pkgname/LICENSE-APACHE"
    install -Dm644 LICENSE-MIT "$pkgdir/usr/share/licenses/$pkgname/LICENSE-MIT"
}
